

## American Communication Systems Discover the Power of Communications™

Discover the Fower of Communications

TO ORDER-VISIT http://www.ameradio.com

## **CT-17**

# COMMUNICATION INTERFACE-V (CI-V) LEVEL CONVERTOR

## INSTRUCTION MANUAL

この取扱説明書は英文、和文共用です。日本語版は11ページからご覧ください。



#### **FOREWORD**

Thank you for purchasing the **CT-17** COMMUNICATION INTERFACE-V (CI-V) LEVEL CONVERTER from ICOM.

Using the CT-17, ICOM CI-V radios can be externally controlled with a personal computer equipped with an EIA standard RS-232C I/O port.

Up to 4 ICOM CI-V radios can be used for frequency, mode, and memory channel control with your program.

A sample program is described below on page 22. Enjoy creating your own programs with the CT-17.

## TABLE OF CONTENTS

1.	CONNECTIONS	1
2.	GENERAL FUNCTIONS	2
	ADDRESS NUMBER	2
	MODEL AND FUNCTION CORRESPOND LIST	3
3.	BASIC PROGRAMS	4
4.	SAMPLE PROGRAM	22
5.	SCHEMATIC DIAGRAM	29

## UNPACKING


Ac	cessories	Q	ty.
1	Mini-plug cables (OPC-017A)		2
2	Power supply cable (OPC-012)		1
<b>(3</b> )	RS-232C cable (OPC-159)		1

## 1. CONNECTIONS

**CAUTION:** Power to the radios and your personal computer must remain OFF before connecting each unit.

- (1) Set the same baud rate for your computer and radio. Refer to your computer instruction manual.
  - The radio is set at the standard ICOM CI-V baud rate of 1200bps.
- (2) Connect the supplied mini-plug cable and RS-232C cable.
  - The inch-type screws are used on the supplied RS-232C cable. If the RS-232C terminal of your computer uses meter-type screws, the supplied meter-type screws must be used to connect the cable.
- (3) Connect the power source to the 9-15V DC IN JACK using the supplied DC cable. The optional BC-25U, BC-25E, BC-26E or BC-27 WALL CHARGER can also be used.
- (4) Turn power to the radios and your personal computer ON.

**NOTE:** If the radio is set near a computer, it may receive computer noise.



## 2. GENERAL FUNCTIONS

#### **ADDRESS NUMBER**

Address numbers of the radios are fixed with the ADDRESS SWITCH or ADDRESS CONNECTOR inside the radios. The ICOM standard address numbers are as follows:

MODEL	ADDRESS NUMBER	MODEL	ADDRESS NUMBER	MODEL	ADDRESS NUMBER
IC-761	1EH (30)	IC-735	04H ( 4)	*IC-471A/E/H	22H (34)
IC-275A/E/H	10H (16)	IC-R7000	08H ( 8)	*IC-1271A/E	24H (36)
IC-475A/E/H	14H (20)	*IC-751A	1CH (28)	*IC-R71A/E/D	1AH (26)
IC-375A	12H (18)	*IC-751	1CH (28)		
IC-575A/E/H	16H (22)	*IC-271A/E/H	20H (32)		

<sup>\*</sup> Address numbers are fixed by the UX-14.

Bracketed figures ( ) are decimals and figures marked with an H are hexadecimals.

#### **FUNCTION**

Each radio can be controlled by your computer as described on p. 3.

# ■ MODEL FUNCTION LIST

FUNCTION	CONTROL	$\begin{array}{c} \text{PROGRAM} \\ \text{No.} \\ \text{(p. 4} \sim 9) \end{array}$	IC-761	IC-275A/E/H IC-375A IC-475A/E/H	IC-735	IC-R7000	IC-751A IC-751 IC-R71A/E/D	IC-271A/E/H IC-471A/E/H IC-1271A/E
				IC-575A/E/H			UX-14 must be used.	st be used.
FREO. SET (transceive mode)	*00	2	Yes	Sə	Yes	Yes	Yes	Yes
MODE SET (transceive mode)	*10	-	Yes	Sə	Yes	Yes	Yes	Yes
BAND EDGE READ	02	4	Yes	Yes	Yes	Yes	Yes	Yes
FREO. READ	03	4	Yes	Yes	Yes	Yes	Yes	Yes
MODE READ	04	4	Yes	Yes	Yes	Yes	Yes	Yes
FREQ. SET	05	2	Yes	Yes	Yes	Yes	Yes	Yes
MQDE SET	90	l	Yes	Yes	Yes	Yes	Yes	Yes
VFO SET	07	1	NOTE	Yes	Yes	No	NOTE	NOTE
MEMORY CH. SET	80	ε	Yes	Yes	Yes	Yes	Yes	Yes
MEMORY WRITE	60	ε	Yes	Yes	Yes	Yes	Yes	Yes
MEMORY VFO	0A	ε	Yes	Yes	Yes	No	Yes	Yes
MEMORY CLEAR	90	8	No	Yes	No	Yes	No	No
OFFSET READ	00	4	No	Yes	No	No	No	Yes
OFFSET WRITE	Q0	2	No	Yes	No	No	No	Yes
SCAN START/STOP	90	3	No	Yes	Š	No	No	No

NOTE: VFO A and B do not change. Only VFO mode is selected.

\* CONTROL COMMAND 00 and 01 ignore each radio's address switch position when the radio address in the program is replaced with 00H.

#### 3. BASIC PROGRAMS

This section explains how to use each control command. Please study the section carefully to thoroughly understand the capabilities of the CI-V System and create your own program using these basic programs.

#### PROGRAM 1 (MODE AND VFO SETTING)

```
10 REM
                  ICOM CI-V Packet Diagram
20 REM ---
30 REM | PR$ | RE$ | TX$ | CN$ | D
40 REM -----
50 REM
60
    PR$=CHR$(&HFE)+CHR$(&HFE)
                                   ' PREAMBLE
70 RE$=CHR$(&H10)
                                   ' TRANSCEIVER ADDRESS
                                                          (IC-275)
80
   TX$=CHR$(&HEO)
                                   ' CONTROLLER
                                                 ADDRESS
   FI$=CHR$(&HFD)
90
                                   ' END
                                           OF
                                                 MESSAGE
100 DEFINT G-Z :CLS
110 OPEN "COM:N81NN" AS #1
120 MD$=" LSB USB A M C W RTTY F M VFO AVFO BE N D"
130 FOR G=1 TO 9 : I=(G-1)*5+1: KEY G, (MID$(MD$, I, 5))
140 NEXT G : KEY 10, "M->VFO"
150 ON KEY GOSUB *A, *B, *C, *D, *E, *F, *G, *H, *I, *J
160 KEY ON
170 LOCATE 25,10 :PRINT "f.1 - f.2 KEY ON " :GOTO 170
180 *A :CN$=CHR$(6):DA$=CHR$(0):GOSUB *DATAOUT:RETURN
190 *B :CN$=CHR$(6):DA$=CHR$(1):GOSUB *DATAOUT:RETURN
200 *C :CN$=CHR$(6):DA$=CHR$(2):GOSUB *DATAOUT:RETURN
210 *D :CN$=CHR$(6):DA$=CHR$(3):GOSUB *DATAOUT:RETURN
220 *E :CN$=CHR$(6):DA$=CHR$(4):GOSUB *DATAOUT:RETURN
230 *F :CN$=CHR$(6):DA$=CHR$(5):GOSUB *DATAOUT:RETURN
240 *G :CN$=CHR$(7):DA$=CHR$(0):GOSUB *DATAOUT:RETURN
250 *H :CN$=CHR$(7):DA$=CHR$(1):GOSUB *DATAOUT:RETURN
260 *J :CN$=CHR$(7):DA$="" :GOSUB *DATAOUT:RETURN
270 *I :KEY OFF :CLOSE :END
1000 *DATAOUT
1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
-1020 PRINT #1, INS$;
1030 RETURN
```

CONTENTS	LINE NUMBER	SAMPLE
Radio address	70	(&H10): IC-275A/E/H (See p. 2)

## ■ PROGRAM 2 (FREQ. OR OFFSET SETTING)

```
ICOM CI-V Packet Diagram
10 REM
20 REM ------
30 REM | PR$ | RE$ | TX$ | CN$ | D A $ | FI$ | 40 REM ------
50 REM
60 PR$=CHR$(&HFE)+CHR$(&HFE)
                                ' PREAMBLE
                                ' TRANSCEIVER ADDRESS (IC-275)
70 RE$=CHR$(&H10)
80 TX$=CHR$(&HEO)
                                 ' CONTROLLER ADDRESS
90 FI$=CHR$(&HFD)
                                 ' END OF
                                              MESSAGE
                                 ' FREQUENCY SET COMMAND
' FREQUENCY DATA
100 CN$=CHR$(&H5)
110 DA$=""
120 DEFINT G-Z : DEFDBL F : DIM CH$(110)
130 FOR M=0 TO 9
140
      FOR N=0 TO 9
150
        CH$(10*M+N) = CHR$(16*M+N)
     NEXT N
160
170 NEXT M
500 OPEN "COM:N81NN" AS #1
510 REM ******* S E T FREQUENCY *******
520 CLS:LOCATE 20,5: INPUT "FREQUENCY INPUT (MHz)"; F
530 F=F+1E-12:IF F<.1 THEN 520
540 F$=RIGHT$("00000"+MID$(STR$(INT(F*1000000#)),2),10)
550 FOR K=1 TO 10 STEP 2
560 DA$=CH$(VAL(MID$(F$,K,2)))+DA$
570 NEXT K
580 GOSUB *DATAOUT :CLOSE :END
1000 *DATAOUT
1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
1020 PRINT #1. INS$:
1030 RETURN
```

CONTENTS	LINE NUMBER	- SAMPLE -
Radio address	70	(&H10): IC-275A/E/H (See p. 2)
Control command	100	(&H5): FREQ. SET Change to (&D) for offset. (See p. 3)
For offset frequency	540	$(F*1000000#)), 2), 10) \rightarrow (F*10000#)), 2), 6)$
	550	TO 10 → TO 6
For IC-735	540	(F*1000000#)), 2), 10) → (F*1000000)), 2), 8)
	550	TO 10 → TO 8

#### ■ PROGRAM 3 (MEMORY AND SCAN OPERATION)

```
ICOM CI-V Packet Diagram
10 REM
30 REM | PR$ | RE$ | TX$ | CN$ | D A $ | FI$ |
40 REM -------
50 REM
60
  PR$=CHR$(&HFE)+CHR$(&HFE) ' PREAMBLE
                               ' TRANSCEIVER ADDRESS (IC-275)
  RE$=CHR$(&H10)
70
                                ' CONTROLLER ADDRESS
80 TX$=CHR$(&HE0)
                                      OF
90 FI$=CHR$(&HFD)
                                ' END
                                             MESSAGE
100 DEFINT G-Z:DIM CH$(110):CLS
110
   FOR M=0 TO 9
120
      FOR N=0 TO 9
130
        CH$(10*M+N) = CHR$(16*M+N)
140
     NEXT N
150 NEXT M
160 OPEN "COM:N81NN" AS #1
170 *MAIN
180 LOCATE 10,5 :PRINT "1 ······ SET MEMORY (VFO --> MEMORY)"
190 LOCATE 10,6 :PRINT "2 ······ SET MEMORY CH "
200 LOCATE 10,7 :PRINT "3 ······ MEMORY WRITE"
210 LOCATE 10,8 :PRINT "4 ······ MEMORY --> VFO WRITE"
220 LOCATE 10,9 :PRINT "5 ····· MEMORY CLEAR"
230 LOCATE 10,10:PRINT "6 ····· SCAN
                                         START"
240 LOCATE 10,11:PRINT "7 ..... SCAN
250 LOCATE 10,12:PRINT "8 · · · · · · · E N D"
260 LOCATE 35,14:PRINT "
270 LOCATE 10,14: INPUT "
                             INPUT COMMAND NO."; I
280 IF I<0 OR I>9 THEN 260
290 IF I=1 THEN CN$=CHR$(8) :DA$="" :GOSUB *DATAOUT :GOTO *MAIN
300 IF I=2 THEN LOCATE 10,16:INPUT "MEMORY CH"; I :CN$=CHR$(8)
:DA$=CH$(I) :GOSUB *DATAOUT :LOCATE 10,16
:PRINT "
                          ":GOTO *MAIN
310 IF I=3 THEN CN$=CHR$(&H9):DA$="" :GOSUB *DATAOUT :GOTO *MAIN
320 IF I=4 THEN CN$=CHR$(&HA):DA$="" :GOSUB *DATAOUT :GOTO *MAIN
330 IF I=5 THEN CN$=CHR$(&HB):DA$="" :GOSUB *DATAOUT :GOTO *MAIN
340 IF I=6 THEN CN$=CHR$(&HE):DA$=CHR$(1):GOSUB)*DATAOUT:GOTO *MAIN
350 IF I=7 THEN CN$=CHR$(&HE):DA$=CHR$(0):GOSUB *DATAOUT:GOTO *MAIN
360 IF I=8 THEN CLOSE :END
370 GOTO *MAIN
1000 *DATAOUT
1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
1020 PRINT #1, INS$;
1030 RETURN
```

CONTENTS	LINE NUMBER	SAMPLE
Radio address	70	(&H10): IC-275A/E/H (See p. 2)

## **■ PROGRAM 4 (PARAMETER READING)**

```
10 REM
                  ICOM CI-V Packet Diagram
20 REM ------
30 REM | PR$ | RE$ | TX$ | CN$ | D A $ | FI$ |
40 REM -----
50 REM
60 PR$=CHR$(&HFE)+CHR$(&HFE)
70 RE$=CHR$(&H10)
80 TX$=CHR$(&HE0)
90 FI$=CHR$(&HFD)
1 END OF MESSAGE
1 COMMAND
50 REM
                                  ' U/L FREQUENCY READ COMMAND
100 CN$=CHR$(&H2)
110 DA$=""
                                   ' FREQUENCY DATA
120 DEFINT A-Z :CLS
130 OPEN "COM: N81NN" AS #1
150 MA$=" LSB USB A M C W RTTY F M "
1000 *DATAOUT
1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
1020 PRINT #1, INS$;
1030 *DATAIN :D$="":Z=0
1040 IF LOC(1)=0 THEN 1040
1050 I$=INPUT$(LOC(1),#1) :L=LEN(I$)
1060 FOR K=1 TO L
1070 P$=MID$(I$,K,1) :M=ASC(P$)
1080 IF P$=CHR$(&HFC) THEN DI$="" :GOTO 1120
1090 IF M<16 THEN X$="0"+HEX$(M) ELSE X$=HEX$(M)
1100 IF M=&HFD THEN DI$=D$ :GOSUB 1500
1110 D$=X$+D$
1120 NEXT K
1130 IF Z>1 THEN 2000
1140 GOTO 1040
1150 IF LEFT$(DI$,8)="FA10E0FE" THEN PRINT " N G ":CLOSE:END
1160 IF LEFT$(DI$,8)="FB10E0FE" THEN PRINT " O K ":CLOSE:END
1170 D$="" :X$=""
1180 RETURN
1500 X=INSTR (DI$,"0210E0FE") : IF X<>0 THEN Z=2
1510 X=INSTR (DI$, "0310E0FE") :IF X<>0 THEN Z=3
1520 X=INSTR (DI$, "0410E0FE") :IF X<>0 THEN Z=4
1530 X=INSTR (DI$, "0C10E0FE") :IF X<>0 THEN Z=5
1540 D$="" :X$=""
1545 PRINT DIS,Z
1550 RETURN
2000 ON (Z-1) GOTO *UPLO, *FRQU, *MODE, *OSET
2010 *UPLO
2020 FH$=LEFT$(DI$,10):FL$=MID$(DI$,13,10)
2030 PRINT VAL(FH$); "Hz -"; VAL(FL$); "Hz"
2040 CLOSE : END
2100 *MODE
2120 IF LEN(DI$)=14 THEN 2150
2130 C=VAL(LEFT$(DI$,2))*5+1:IF C>37 THEN MO$="- - - -" :GOTO 2160
2140 MO$=MID$(MA$,C,5):GOTO 2160
2150 IF LEFT$(DI$,4)="0005" THEN MO$="SSB(R7000)" ELSE MO$=LEFT$(DI$,4)
2160 PRINT MO$ :CLOSE :END
2200 *FRQU
2210 IF LEFT$(DI$,2)="FF" THEN F$="- - - - - - - ":GOTO 2230
2220 F$=LEFT$(DI$,10) :F#=VAL(F$)/1000000#
2230 PRINT F#;"MHz" :CLOSE :END
2300 *OSET
2310 F$=LEFT$(DI$,6) :F#=VAL(F$)/10000#
2320 PRINT F#;"MHz" :CLOSE :END
```

	CONTENTS	LINE NUMBER	SAMPLE
	Radio address	70	(&H10): IC-275A/E/H (See p. 2)
		1150, 1160	"FA <u>10</u> E0FE": IC-275A/E/H (See p. 3)
-		1500~1530	"02 <u>10</u> E0FE" : IC-275A/E/H (See p. 3)
	Control command	100	(&H2): Band edge reading. (&H3), (&H4) or (&HC) is acceptable. (See p. 3)
	For IC-735	2020	(DI\$, 10) → (DI\$, 8) (DI\$, 13, 10) → (DI\$, 11, 8)
		2220	(DI\$, 10) → (DI\$, 8)

NOTE 1: Line numbers  $1000 \sim 1180$  are a confirmation program that can also be used for PROGRAMS 1, 2 and 3 (line number  $1000 \sim 1030$  on PROGRAMS 1, 2 or 3). This program displays "OK" or "NG" on the computer CRT DISPLAY to check the command operation.

When this program is used, GOSUB 1500 in line number 1100 must be replaced with GOSUB 1150, and control codes 00 and 01 cannot be accepted.

NOTE 2: This program does not read Narrow mode (i.e., CW-Narrow, RTTY-Narrow). When the radio is in such a mode the following mode codes appear on the computer CRT DISPLAY.

WIDTH CODE	WIDTH	MODE CODE	MODE	MODE CODE	MODE						
01	Wide (Normal)	00	LSB	03	CW						
02	Narrow	01	USB	04	RTTY						
03	Super-narrow	02	AM	05	FM						
i.e., CW-Na	rrow : 0203				i.e., CW-Narrow: 0203						

NOTE 3: If the wrong address number is entered in this program, it is possible that an infinite loop is made, so the computer does not answer. In this case, the program should be stopped (i.e., push STOP KEY).

#### Continuous parameter reading

When making the following modifications for PROGRAM 4 the frequency and mode are shown on the computer CRT DISPLAY each time they change.

- 1. Erase line numbers  $60 \sim 110$ .
- 2. Erase line numbers  $1000 \sim 1020$ .
- 3. Erase line numbers  $1100 \sim 2320$ .
- 4. Turns ON the transceiver flag switch inside the radio.
- 5. Inputs the following program.

```
1100 IF M=&HFD THEN DI$=D$ :GOSUB 3000
1110 D$=X$+D$
1120 NEXT K
1140 GOTO 1040
3000 X=INSTR (DI$,"000200FE") : IF X<>0 THEN 3100
3020 X=INSTR (DI$,"010200FE") :IF X<>0 THEN 3200
3040 D$="" :X$=""
3050 RETURN
3100 F$=LEFT$(DI$,10):F#=VAL(F$)/1E+06
3140 PRINT USING "######.#####MHz":F#
3150 GOTO 3040
3200 IF LEN(DI$)=14 THEN 3230
3210 C=VAL(LEFT$(DI$,2))*5+1 :IF C>37 THEN MO$="----" :GOTO 3260
3220 MO$=MID$(MA$,C,5):GOTO 3260
3230 IF LEFT$(DI$,4)="0005" THEN MO$="SSB(R7000)"
3235 IF LEFT$(DI$,4)="0205" THEN MO$="FM-N"
3240 IF LEFT$(DI$,4)="0204" THEN MO$="RTTY-N"
3245 IF LEFT$(DI$,4)="0203" THEN MO$=" C W-N"
3260 '
3270 PRINT MO$ :GOTO 3040
```

CONTENTS	LINE NUMBER	SAMPLE
Radio address	3000	(DI\$, "00 <u>02</u> 00FE"): Change to required address number. (See p. 2)
	3020	(DI\$, "01 <u>02</u> 00FE"): Change to required address number. (See p. 2)
For IC-735	3100	(DI\$, 10) → (DI\$, 8)

**NOTE:** When the frequency or mode is changed quickly the computer may not read each parameter.

## ICOM INCORPORATED

1-6-19, KAMIKURATSUKURI, HIRANO-KU, OSAKA 547, JAPAN

Printed in Japan

## **CT-17**

# 取扱説明書

# 日本語編

## 目 次

はじめに12	<del>)</del>
付属品12	<u>)</u> .
CT-17の接続方法	}
CI-Vの機能14	1
アドレスについて14	ļ
機能一覧表15	;
基本プログラムと操作16	
■プログラムNo.1(MODE, VFOの切換え)······16	3
■プログラムNo.2(周波数、オフセット周波数の設定)17	7
■プログラムNo.3(MEMORYおよびSCANの設定)18	}
■プログラムNo.4	
①データの読み込み19	}
②データの連続読み込み21	
SAMPLE PROGRAM······22	2~28
回路図29	}

## はじめに

この度はアイコム製品をお買い上げいただき、まことにありがとうございます。 アイコムのトランシーバー(固定機)には、マイクロプロセッサーが搭載されており、これにより外部からのリモートコントロールを可能にしています。

- このシステムをICOM COMMUNICATION INTERFACE-V(CI-V)システムと呼び、 パーソナルコンピューターなどの外部機器とシリアルバスラインで接続することが できます。

CT-17は、EIA規格RS-232Cシリアルポートを持つパーソナルコンピューターと、トランシーバーを接続するための電圧レベルコンバーターです。

ご使用の際は、この取扱説明書をよくお読みいただき、パーソナルコンピューターのプログラムで、多彩なコントロールをお楽しみください。


## 付 属 品

OPC-017Aミニプラグケーブル	. 5
OPC-012電源ケーブル	. 1
OPC-159 RS-232ケーブル	. 1

## CT-17の接続方法

## ご注意

- ●パーソナルコンピューターおよびトランシーバー接続の際は、必らず 電源を切ってから接続してください。
- ●伝送速度(ボーレート)について パーソナルコンピューターとトランシーバー間の伝送速度は、トランシーバー内蔵のディップスイッチで設定することができますが、CI-V標準伝送速度として1,200bpsに初期設定されています。 パーソナルコンピューター側のボーレート設定も忘れずに行なってください。
- ●RS-232Cケーブルにはインチネジが付いています。 手持のパーソナルコンピューターのRS-232Cターミナルのネジがミリネジを使用している場合はケーブルのコンピューター側のネジを付属のミリネジと交換してください。
- (1) 接続は下図のように接続してください。 CT-17には4個の入出力ポートがありますので、トランシーバーを最大4台まで 接続できます。
- (2) CT-17の電源はDC9~15Vの範囲です。付属のDCケーブルで接続してください。 オプションのBC-25(AC100V→DC13.8Vコンバーター)も接続できますのでご利 用ください。



, ※コンピューターとトランシーバーを近くで使用すると、コンピューターのノー イズがトランシーバーに入ることがありますのでご注意ください。

## CI-Vの機能

アイコムトランシーバーにはCI-V(シリアルI/Oポート)システム搭載機とCI-W(パラレルI/Oポート)システム搭載機があります。

CI-IVシステムのトランシーバーでも、CI-IV/CI-VコンバーターUX-14(オプション)を接続することにより、CI-Vシステムに準じた外部コントロールができます。 トランシーバーの機種により、多少の相異はありますが、右記一覧表の機能をコントロールすることができます。

#### 1.アドレスについて

- CI-Vシステムでは、コントローラーに複数台のトランシーバーを接続してコントロールするため、それぞれの機器に固有のデバイスアドレスを設けています。アドレスの設定は、トランシーバーの内部にあるディップスイッチ(CI-V搭載機)またはUX-14の基板上にあるディップスイッチ(CI-V搭載機)で任意に設定が可能ですが、アイコムCI-V標準アドレスとして次のように登録されています。

なお、コントローラーのアドレスはEO(16進)を設定してください。

機種	アドレス	機  種	アドレス
IC-731/S	04H	IC-750/S, 750A/AS	1CH
IC-R7000	08H	IC-760	1EH
IC-275/D	10H	IC-271	20H
IC-375/D	14H	IC-371	22H
IC-575	16H	IC-1271	24H
IC-R71	1AH		

アドレスの設定にはOHおよびFOH~FFH(0および240~255)は使用しないでください。

※CI-VおよびCI-Ⅳ搭載機の区別は15ページの表を参照してください。

機能一覧表

	コマン	参照		CI-V	CI-V搭載機		CI-IV搭載機	苔載機
额	ド (命令) 16篇	ロム	IC-760	IC-275/D IC-375/D IC-575	IC-731/S	IC-R7000	IC-750/S IC-751A/AS IC-R71	IC-271/D IC-371/D IC-1271
周波数データの転送(トランシープON)	00	2	0	0	0	0	0	
MODEデータの転送(トランシープON)	01	_	0		0		0	0
バンドエッジ周波数の読み込み	02	4	0	0	0	0		0
表示周波数の読み込み	03	4	0	0	0	0		0
表示MODEの読み込み	04	4	0	0	0	0	0	0
周波数データの転送	05	2			0	0	0	0
MODEデータの転送	90	_	0	0	0	0		ß.
VFO AまたはBのセット	07		4		0	×	◁	◁
MEMO-CHOセット	08	3	0		0	0	0	0
MEMORYへ書き込み	60	3		0	0	0	0	0
MEMO/VFO切換え	OA	3	$\circ$	0	0	×	0	
MEMORY5UF	90	m	×	0	×	0	×	×
オフセット周波数の読み込み	20	4	×	0	×	×	×	0
オフセット周波数の書き込み	00	2	×	0	× .	×	×	0
SCAN START/STOP	OE	m	×	0	×	×	×	×

※AEIはVFO A/Bの切換えはできませんが、MEMO状態からVFOモード(元のVFO)に戻すことができます。

## 基本プログラムと操作

パーソナルコンピューターを利用して、トランシーバーを外部コントロールする 各種の命令(機能)は、前ページの一覧表のとおりです。

これらの機能を使って、簡単なプログラムを下記に掲載します。

このプログラムは、パーソナルコンピューターからデータを与えるもの、トランシーバーからデータを引き出すもの、トランシーバーを動作させるものなどに分類し、命令部やアドレス部などの数値を変えることにより、種々の機能を発揮させるようにしています。

ボーレート、アドレスなどの設定および、機器の接続に間違いがないかをよく確認したうえで、下記プログラムを操作してください。

※例題は、NEC N-88BASICを使用しています。

## ■プログラム No.1(MODE, VFOの切換え)

```
ICOM CI-V Packet Diagram
10 REM
20 REM -
                   | RE$ | TX$ | CN$ |
30 REM |
            PR$
                                            D
40 REM --
50 REM
60
   PR$=CHR$(&HFE)+CHR$(&HFE)
                                   ' PREAMBLE
70
   RE$=CHR$(&H10)
                                   ' TRANSCEIVER ADDRESS
                                                          (IC-275)
                                   ' CONTROLLER
80
    TX$=CHR$(&HEO)
                                                 ADDRESS
90
    FI$=CHR$(&HFD)
                                   ' END
                                           OF
                                                 MESSAGE
100
    DEFINT G-Z :CLS
110 OPEN "COM: N81NN" AS #1
120 MD$=" LSB USB A M C W RTTY F M VFO AVFO BE N D"
130 FOR G=1 TO 9 : I=(G-1)*5+1 : KEY G. (MID$ (MD$, I.5))
140 NEXT G : KEY 10, "M->VFO"
150 ON KEY GOSUB *A, *B, *C, *D, *E, *F, *G, *H, *I, *J
160 KEY ON
170 LOCATE 25,10 :PRINT "f.1 - f.2 KEY ON " :GOTO 170
180 *A :CN$=CHR$(6):DA$=CHR$(0):GOSUB *DATAOUT:RETURN
190 *B :CN$=CHR$(6):DA$=CHR$(1):GOSUB *DATAOUT:RETURN
200 *C :CN$=CHR$(6):DA$=CHR$(2):GOSUB *DATAOUT:RETURN
210 *D :CN$=CHR$(6):DA$=CHR$(3):GOSUB *DATAOUT:RETURN
220 *E :CN$=CHR$(6):DA$=CHR$(4):GOSUB *DATAOUT:RETURN
230 *F :CN$=CHR$(6):DA$=CHR$(5):GOSUB *DATAOUT:RETURN
240 *G :CN$=CHR$(7):DA$=CHR$(0):GOSUB *DATAOUT:RETURN
250 *H :CN$=CHR$(7):DA$=CHR$(1):GOSUB *DATAOUT:RETURN
260 *J :CN$=CHR$(7):DA$="" :GOSUB *DATAOUT:RETURN
270 *I :KEY OFF :CLOSE :END
1000 *DATAOUT
1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
1020 PRINT #1, INS$;
1030 RETURN
```

項		文番号	プログラムの修正内容
トランシーバ- の変更	ーアドレス	70	(&H10):IC-275のアドレスです。接続機種のアドレスに変 更してください。

注:VFO A/Bの切換えは機種により動作しないものがあります。

## ■プログラム No.2(周波数, オフセット周波数の設定)

```
10 REM
                 ICOM CI-V Packet Diagram
 20 REM -----
 30 REM | PR$ | RE$ | TX$ | CN$ | D A $ | FI$ |
 40 REM ------
 50 REM
 60
   PR$=CHR$(&HFE)+CHR$(&HFE)
                                ' PREAMBLE
 70 RE$=CHR$(&H10)
                                'TRANSCEIVER ADDRESS (IC-275)
                               ' CONTROLLER ADDRESS
 80 TX$=CHR$(&HEO)
 90 FI$=CHR$(&HFD)
                                ' END OF MESSAGE
                                ' FREQUENCY SET COMMAND
' FREQUENCY DATA
 100 CN$=CHR$(&H5)
 110 DA$=""
 120 DEFINT G-Z : DEFDBL F : DIM CH$(110)
    FOR M=0 TO 9
 130
 140
       FOR N=0 TO 9
 150
         CH$(10*M+N) = CHR$(16*M+N)
 160
 170 NEXT M
 500 OPEN "COM: N81NN" AS #1
      REM ******* S E T FREQUENCY *******
 510
 520 CLS:LOCATE 20,5: INPUT "FREQUENCY INPUT (MHz)"; F
 530 F=F+1E-12:IF F<.1 THEN 520
 540 F$=RIGHT$("00000"+MID$(STR$(INT(F*1000000#)),2),10)
 550 FOR K=1 TO 10 STEP 2
 560 DA$=CH$(VAL(MID$(F$,K,2)))+DA$
₹ 570 NEXT K
 580 GOSUB *DATAOUT :CLOSE :END
 1000 *DATAOUT
 1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
 1020 PRINT #1, INS$;
 1030 RETURN
```

項目	文番号	プログラムの修正内容
トランシーバーアドレス の変更	70	(&H10):IC-275のアドレスです。接続機種のアドレスに変 更してください。
コマンド変更	100	(&H5):周波数の書き込み命令です。 オフセットの場合は(&HD)にします。
周波数書き込みを オフセット書き込みに	540	(F×1000000#)),2),10)を (F×10000#)),2,6)に変更してください。
	550	TO10をTO6に変更してください。
※IC-731の場合	540	(F×1000000#)),2,10)を (F×1000000#)),2,8)に変更してください。
	550	TO10をTO8に変更してください。

※IC-731の周波数データは8バイト(桁)です。 他の機種は10バイト(桁)になっています。

## ■プログラム No.3(MEMORYおよびSCANの設定)

```
ICOM CI-V Packet Diagram
10 REM
20 REM -----
30 REM | PR$ | RE$ | TX$ | CN$ | D A $ | FI$ |
40 REM -----
50 REM
60 PR$=CHR$(&HFE)+CHR$(&HFE)
                               ' PREAMBLE
70 RE$=CHR$(&H10)
                                ' TRANSCEIVER ADDRESS (IC-275)
                                ' CONTROLLER ADDRESS
80 TX$=CHR$(&HE0)
                                ' END OF MESSAGE
90 FI$=CHR$(&HFD)
100 DEFINT G-Z : DIM CH$(110) : CLS
110 FOR M=0 TO 9
120 FOR N=0 TO 9
130
       CH$(10*M+N)=CHR$(16*M+N)
140
150 NEXT M
160 OPEN "COM:N81NN" AS #1
170 *MAIN
180 LOCATE 10,5 :PRINT "1 ······ SET MEMORY (VFO --> MEMORY)"
190 LOCATE 10,6 :PRINT "2 ······ SET MEMORY CH "
200 LOCATE 10,7 :PRINT "3 ····· MEMORY WRITE"
210 LOCATE 10,8 :PRINT "4 ······ MEMORY --> VFO WRITE"
220 LOCATE 10,9 :PRINT "5 ······ MEMORY CLEAR"
230 LOCATE 10,10:PRINT "6 ····· SCAN
                                        START"
240 LOCATE 10,11:PRINT "7 ······ SCAN
                                          STOP"
250 LOCATE 10,12:PRINT "8 .... E N D"
260 LOCATE 35,14:PRINT "
270 LOCATE 10,14:INPUT "
                            INPUT COMMAND NO.":I
280 IF I O OR I > 9 THEN 260
290 IF I=1 THEN CN$=CHR$(8) :DA$="" :GOSUB *DATAOUT :GOTO *MAIN
300 IF I=2 THEN LOCATE 10,16:INPUT "MEMORY CH"; I :CN$=CHR$(8)
:DA$=CH$(I) :GOSUB *DATAOUT :LOCATE 10,16
:PRINT "
                           ":GOTO *MAIN
310 IF I=3 THEN CN$=CHR$(&H9):DA$="" :GOSUB *DATAOUT :GOTO *MAIN
320 IF I=4 THEN CN$=CHR$(&HA):DA$="" :GOSUB *DATAOUT :GOTO *MAIN
330 IF I=5 THEN CN$=CHR$(&HB):DA$="" :GOSUB *DATAOUT :GOTO *MAIN
340 IF I=6 THEN CN$=CHR$(&HE):DA$=CHR$(1):GOSUB)*DATAOUT:GOTO *MAIN
350 IF I=7 THEN CN$=CHR$(8HE):DA$=CHR$(0):GOSUB *DATAOUT:GOTO *MAIN 360 IF I=8 THEN CLOSE :END
370 GOTO *MAIN
1000 *DATAOUT
1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
1020 PRINT #1, INS$;
1030 RETURN
```

項		文番号	プログラムの修正内容
トランシーバ の変更	ーアドレス	70	(&H10):IC-275のアドレスです。接続機種のアドレスに変 更してください。

注:SCAN START/STOPは機種により動作しないものがあります。

#### ■プログラム No.4

#### ①データの読み込み

```
ICOM CI-V Packet Diagram
 10 REM
 30 REM | PR$ | RE$ | TX$ | CN$ | D A $ | FI$ |
 40 REM -----
 50 REM
 50 REM
60 PR$=CHR$(&HFE)+CHR$(&HFE)
' PREAMBLE
' TRANSCEIVER ADDRESS (IC-275)
                                ' CONTROLLER ADDRESS
 80 TX$=CHR$(8HE0)
                                 ' END OF
 90 FI$=CHR$(&HFD)
                                              MESSAGE
                                 ' U/L FREQUENCY READ COMMAND
 100 CN$=CHR$(&H2)
                                 ' FREQUENCY
 110 DA$=""
                                              DATA
 120 DEFINT A-Z :CLS
 130 OPEN "COM:N81NN" AS #1
 150 MA$=" LSB USB A M C W RTTY F M "
 1000 *DATAOUT
 1010 INS$=PR$+RE$+TX$+CN$+DA$+FI$
p. 1020 PRINT #1, INS$;
 1030 *DATAIN :D$="":Z=0
 1040 IF LOC(1)=0 THEN 1040
 1050 I = INPUT $ (LOC(1), #1) : L=LEN(I$)
 1060 FOR K=1 TO L
 1070 P$=MID$(I$,K,1) :M=ASC(P$)
 1080 IF P$=CHR$(&HFC) THEN DI$="" :GOTO 1120
 1090 IF M<16 THEN X$="0"+HEX$(M) ELSE X$=HEX$(M)
 1100 IF M=&HFD THEN DI$=D$ :GOSUB 1500
 1110 D$=X$+D$
 1120 NEXT K
 1130 IF Z>1 THEN 2000
 1140 GOTO 1040
 1150 IF LEFT$(DI$,8)="FA10E0FE" THEN PRINT " N G ":CLOSE:END
 1160 IF LEFT$(DI$,8)="FB10E0FE" THEN PRINT " O K ":CLOSE:END
 1170 D$="" :X$=""
 1180 RETURN
 1500 X=INSTR (DI$,"0210E0FE") : IF X<>0 THEN Z=2
 1510 X=INSTR (DI$, "0310E0FE") : IF X<>0 THEN Z=3
 1520 X=INSTR (DI$,"0410E0FE") : IF X<>0 THEN Z=4
 1530 X=INSTR (DI$, "OC10E0FE") : IF X<>0 THEN Z=5
 1540 D$="" :X$=""
 1545 PRINT DI$,Z
 1550 RETURN
 2000 ON (Z-1) GOTO *UPLO, *FRQU, *MODE, *OSET
 2010 *UPLO
 2020 FH$=LEFT$(DI$,10):FL$=MID$(DI$,13,10)
 2030 PRINT VAL(FH$); "Hz -"; VAL(FL$); "Hz"
 2040 CLOSE : END
 2100 *MODE
 2120 IF LEN(DI$)=14 THEN 2150
 2130 C=VAL(LEFT$(DI$,2))*5+1:IF C>37 THEN MO$="- - - -" :GOTO 2160
 2140 MO$=MID$(MA$,C,5):GOTO 2160
 2150 IF LEFT$(DI$,4)="0005" THEN MO$="SSB(R7000)" ELSE MO$=LEFT$(DI$,4)
 2160 PRINT MO$ :CLOSE :END
 2200 *FRQU
 2210 IF LEFT$(DI$,2)="FF" THEN F$="- - - - - - - ":GOTO 2230
 2220 F$=LEFT$(DI$.10) :F#=VAL(F$)/1000000#
 2230 PRINT F#;"MHz" :CLOSE :END
 2300 *OSET
 2310 F$=LEFT$(DI$,6) :F#=VAL(F$)/10000#
 2320 PRINT F#;"MHz" :CLOSE :END
```

項	8	文番号	プログラム修正の内容
トランシーバ- の変更	-アドレス	70	(&H10):IC-275のアドレスです。接続機種のアドレスに変 更してください。
トランシーバ- の変更	-アドレス	1150 \	"FA10EOFE":左記データ文の中の10がアドレスです。指定 (FA10EOFE" 機器のアドレスに変更してください。
コマンドの変動	更 ·	100	(&H2):バンドエッジ周波数の読み込みです。 (&H3)周波数読み込み に変更することが (&H4)MODE読み込み たきます。 (&HC)オフセット読み込み たきます。
※IC-731の場	合	2020 2220	(DI\$, 10)を(DI\$, 8)に (DI\$, 13, 10)を(DI\$, 11, 8)に (DI\$, 10)を(DI\$, 8)に変更してください。

※IC-731の周波数データ部は8バイト(桁)です。

注1 文番号1000~1180は確認用プログラムで、トランシーバーはコマンドメッセージに対して"OK"または"NG"を返送します。

この処理ルーチンを例題プログラムNo.1, 2, 3の文番号1000~1030の代わりに入れますと、"OK""NG"の確認ができます。

なお、文番号1100のGOSUB1500を1150に変更してください。

ただしこの場合、コマンド&HOと&H1(周波数およびMODEデータの連続読み込み)は使用できません。

注2 トランシーバーのアドレスが一致しない(アドレス指定の誤りまたはアドレス スイッチのセットミス)場合は、トランシーバーからの応答がありません。例 題プログラムはこのとき、読み込み待ちの状態を続けます。

この場合はSTOPまたはRESETキーを押しますとプログラムはEND処理まで進みます。

注3 このプログラムにはCWナロー等のナローモードの表示はできません。もしトランシーバーがナローモードの場合は下記の表に従ったコードが現われます。

帯域コード	帯域幅	モードコード	モ ー ド
01	ワイド(ノーマル)	00	LSB
02	ナロー 極ナロー	01	USB
03	極ナロー	02	AM
		03	CW
		04	RTTY
		05	FM

例:CWナローの場合 0203 となります。

注4 MEMO-CHの読み込み時、ブランクチャンネルの場合は、機種によりブランクデータの場合と周波数データの場合があります。

#### ②データの連続読み込み

トランシーバー内部のトランシーブスイッチをONにしますと、下記のような変更により、特定のトランシーバーの周波数やMODEを変えるたびに、そのデータが送られて来ます。

(この機能はトランシーバ同志のトランシーブ機能を利用しています。)

```
1100 IF M=&HFD THEN DI$=D$ :GOSUB 3000
1110 D$=X$+D$
1120 NEXT K
1140 GOTO 1040
3000 X=INSTR (DI$,"000200FE") :IF X<>0 THEN 3100
3020 X=INSTR (DI$,"010200FE") : IF X<>0 THEN 3200
3040 D$="" :X$=""
3050 RETURN
3100 F$=LEFT$(DI$,10):F#=VAL(F$)/1E+06
3140 PRINT USING "##############Hz":F#
3150 GOTO 3040
3200 IF LEN(DI$)=14 THEN 3230
3210 C=VAL(LEFT$(DI$,2))*5+1 :IF C>37 THEN MO$="----" :GOTO 3260
3220 MO$=MID$(MA$,C,5):GOTO 3260
3230 IF LEFT$(DI$,4)="0005" THEN MO$="SSB(R7000)"
3235 IF LEFT$(DI$,4)="0205" THEN MO$="FM-N"
3240 IF LEFT$(DI$,4)="0204" THEN MO$="RTTY-N"
3245 IF LEFT$(DI$,4)="0203" THEN MO$=" C W-N"
3260 '
3270 PRINT MO$ :GOTO 3040
```

#### ●プログラムNo.4の変更

項		文番号	プログラム修正の内容
削	除	60  110	文番号60~110を消去してください。
削	除	1000 \$ 1020	文番号1000~1020を消去してください。
削	除	1110 \$ 2320	文番号1110〜2320を消去してください。
トランシ	ーバーアドレス	3000	(DI\$,"000200FE"): 02がアドレス部です。指定の機器アドレスに変更してくださ い。
の変更		3020	(DI\$,"010200FE"): 02がアドレス部です。指定の機器アドレスに変更してくださ い。

注:トランシーバーの周波数やモードを速く変化させた場合、コンピューターが読みきれないことがあります。

:IC-731/Sの場合、文番号3100の(DI\$10)を(DI\$8)に変更してください。

#### 4. SAMPLE PROGRAM

This sample program is developed for use with the NEC N88-BASIC language for NEC PC-9800 series computer. Make modifications as necessary to suit your personal computer using the program.

#### **CAUTIONS:**

- 1. Use controller address E0.
- 2. Do not set address switch inside radio for address numbers 0H and F0H  $\sim$  FFH (0 and 240  $\sim$  255).
- 3. When a blank status memory channel is selected and appears on the radio display, the memory channel is blank even if the computer shows a previously sent frequency on the computer display. This discrepancy may occur with some radios.
- 4. This program is made for most radios using 10 byte for frequency reading and setting (including band edge data). The IC-735, however, uses 8 byte, so refer to PROGRAMS 2 and 4 in the BASIC PROGRAMS section when programming for the IC-735.

## SAMPLE PROGRAM

おわりにCI-Vシステムのすべての機能をコントロールするプログラム例を掲載します。

BASIC言語はもとより、C言語、アッセンブラ、マシン語など、お手持ちのパーソナルコンピューターに合わせてプログラムを組むことも可能です。

下記例のBASICプログラムで、CI-Vをご理解いただき、さらに発展させた貴方独自のプログラミングをお楽しみください。

```
10 '
      ************
20 '
             ICOM CI-V
30 '
            CT-17 REMOTE CONTROL
                                       **
40 '
      **
             86/06/06
                        Ver. 1.0
                                       **
50 '
      ** Copyright(C) by ICOM INC.
60 '
      ***********
70 '
                   ICOM CI-V Packet Diagram
80
90 '
100 '
        ! PR$ ! RE$ ! TX$ ! CN$ ! D A $ ! FI$ !
110 '
                   ______
120 '
                                  ' PREAMBLE
130 PR$=CHR$(&HFE)+CHR$(&HFE)
                                   ' TRANSCEIVER ADDRESS : IC-R7000
140 RE$=CHR$(&H8)
                                   ' CONTROLLER ADDRESS : 0E0H
150 TX$=CHR$(&HEO)
                                   ' CONTROL CODE
160 CN$=""
                                   ' DATA
170 DA$=""
180 FI$=CHR$(&HFD)
                                   ' END OF MESSAGE
190
200
    MDA$="LSB USB A M C W RTTYF M "
210 WIDTH 80,25 :CONSOLE 20,5,0,0
220 DIM CH$(110),FQ$(110),MO$(110),MM$(110),FF$(110)
230 DEFDBL B: DEFINT C-Z
240 CX=0: CY=20: LY1=1: LY2=1: LY3=1: LY4=1: LY5=1
250 B$="
                          ": B1$=""
260 ME$=CHR$(8H1B)+CHR$(8H4B)+CHR$(8H22)+CHR$(8H2A)+CHR$(8H1B)+CHR$(8H41)
270 ON STOP GOSUB 3410
280 STOP ON
290 FOR M=0 TO 9
300
      FOR N=0 TO 9
310
        CH$(10*M+N)=CHR$(16*M+N)
320
      NEXT N
330 NEXT M
340 OPEN "COM:N81NN" AS #1
350 ON COM GOSUB 2730
360 GOSUB 3260
370 LOCATE 6,9 :PRINT "M O D E "
380 LOCATE 6,11 :PRINT "FREQUENCY"
390 '
400 '
          MENU ROUTINE
410 '
420 LOCATE 50,1 : PRINT "◆ SET FREQUENCY (TRANSCIEVE) "
430 LOCATE 50,2 : PRINT "♦ SET MODE (TRANSCIEVE) "
440 LOCATE 50,3 : PRINT "♦ READ UPPER/LOWER FREQUENCY "
450 LOCATE 50,4 : PRINT "♦ READ FREQUENCY "
460 LOCATE 50,5 : PRINT "♦ READ MODE "
470 LOCATE 50,6 : PRINT "♦ SET FREQUENCY "
480 LOCATE 50,7 : PRINT "♦ SET MODE "
490 LOCATE 50,8 : PRINT "♦ SET VFO "
500 LOCATE 50,9 : PRINT "♦ SET MEMORY "
510 LOCATE 50,10: PRINT "♦ MEMORY WRITE "
520 LOCATE 50,11: PRINT "♦ VFO WRITE (M-CH -> VFO) "
530 LOCATE 50,12: PRINT "♦ MEMORY CLEAR "
540 LOCATE 50,13: PRINT "◆ READ OFFSET FREQUENCY "
550 LOCATE 50,14: PRINT "♦ SET OFFSET FREQUENCY "
560 LOCATE 50,15: PRINT "♦ SCAN CONTROL"
570 COM ON
580 LOCATE 50, LY2
590 COLOR@(50,LY2)-(50,LY2),6
600 A$= INKEY$
610 IF A$="" THEN GOTO 600
620 N=15: J=50: LY=LY2
630 GOSUB 3050
640 LY2=LY
650 ON K GOTO 600,590,660,600
660 COLOR@(50,LY2)-(50,LY2),0
670 ON LY2 GOTO 770,1270,710,710,710,770,1270,1570,1810,710,710,710 ,710,2290
.2400
680 '
690 '
          ANSWER COMMAND
```

700 '

```
710 DA$=""
720 GOSUB 2620
730 GOTO 590
740 '
750
          FREQUENCY WRITE
760 '
770 COM STOP
780 FQ=1
790 LOCATE 25,4: PRINT "FREQUENCY INPUT (MHz)"
800 LOCATE 29,6: PRINT """
810 LOCATE 25,8: PRINT ME$;" key then return to MENU"
820 COLOR@(29,6)-(29,6),6
830 COM ON
840 L=10: M=6
850 A$= INKEY$
860 IF A$="" THEN GOTO 850
870 IF INSTR ("0123456789."+CHR$(13)+CHR$(28),A$)=0 THEN GOTO 850
880 IF A$=CHR$(13) THEN GOTO 1080
890 IF A$=CHR$(28) THEN GOSUB 2960: GOTO 590
900 IF K=1 THEN 1000
910 IF A$="." THEN 970
920 FOR N=1 TO 3
930 MID$(B$,N,1)=MID$(B$,N+1,1)
940 NEXT N
950 MID$(B$,4,1)=A$
960 GOTO 1040
970 MID$(B$,5,1)="."
980 K=1
990 GOTO 1040
1000 IF M=12 THEN GOTO 1040
1010 MID$(B$,M,1) =A$
1020 M=M+1
1030 GOTO 1040
1040 COM STOP
1050 LOCATE 30,6: PRINT B$
1060 COM ON
1070 GOTO 850
1080 MID$(B$,M,6)="000000"
1090 FOR N=L TO 6 STEP -2
1100 B1$=MID$(B$,N,2)
1110 F=VAL(B1$)
1120 DA$=DA$+CH$(F)
1130 NEXT N
1140 FOR N=3 TO 1 STEP -2
1150 B1$=MID$(B$,N,2)
1160 F=VAL(B1$)
1170 DA$=DA$+CH$(F)
1180 IF L=8 THEN GOTO 1200
1190 NEXT N
1200 B$="
1210 COLOR@(26,6)-(26,6),0
1220 GOSUB 2620: GOSUB 2960
1230 IF FQ=1 THEN GOTO 770 ELSE 2290
1240
1250
           MODE WRITE
1260
1270 COM STOP
1280 LOCATE 25,1 : PRINT "♦ LSB"
1290 LOCATE 25,2 : PRINT "♦ USB"
1300 LOCATE 25,3 : PRINT "♦ AM"
1310 LOCATE 25,4 : PRINT "♦ CW"
1320 LOCATE 25,5 : PRINT, "◆ RTTY"
1330 LOCATE 25,6 : PRINT "→ FM"
1340 LOCATE 25,7 : PRINT "◆ FM-n"
1350 LOCATE 25,8 : PRINT "♦ SSB (IC-R7000 ONLY)"
1360 LOCATE 25,10: PRINT ME$;" key then return to MENU"
1370 COM ON
1380 LOCATE 25,LY1
1390 COLOR@(25,LY1)-(25,LY1),6
1400 A$= INKEY$
1410 IF A$="" THEN GOTO 1400
```


```
1420 N=9: J=25: LY=LY1
1430 GOSUB 3050
1440 LY1=LY
1450 ON K GOTO 1400,1390,1460,1520
1460 COLOR@(25,LY1)-(25,LY1),0
1470 C=LY1-1
1480 ON LY1 GOTO 1500,1500,1500,1500,1500,1500,1510,1490
1490 DA$=CH$(5)+CH$(0):GOSUB 2620: GOTO 1390
1500 DA$=CH$(C):GOSUB 2620: GOTO 1390
1510 DA$=CH$(5)+CH$(2):GOSUB 2620: GOTO 1390
1520 GOSUB 2960
1530 GOTO 590
1540
1550
           VFO CHANGING
1560
1570 COM STOP
1580 LOCATE 25,1 : PRINT "♦ VFO A"
1590 LOCATE 25,2 : PRINT "♦ VFO B"
1600 LOCATE 25,3 : PRINT "♦ M-CH MODE -> VFO MODE"
1610 LOCATE 25,5 : PRINT ME$;" key then return to MENU"
1620 COM ON
1630 LOCATE 25, LY3
1640 COLOR@(25,LY3)-(25,LY3),6
1650 A$=INKEY$
1660 IF A$="" THEN GOTO 1650
1670 N=3: J=25: LY=LY3
1680 GOSUB 3050
1690 LY3=LY
1700 ON K GOTO 1650,1640,1710,1760
1710 COLOR@(25,LY3)-(25,LY3),0
1720 C=LY3-1
1730 ON LY3 GOTO 1740,1740,1750
1740 DA$=CH$(C): GOSUB 2620: GOTO 1640
1750 DA$="": GOSUB 2620: GOTO 1640
1760 GOSUB 2960
1770 GOTO 590
1780
           MEMORY CALL
1790 '
1800 '
1810 COM STOP
1820 LOCATE 25,1 : PRINT "◆ VFO MODE -> M-CH MODE"
1830 LOCATE 25,2 : PRINT "♦ CH INPUT"
1840 LOCATE 25,8 : PRINT ME$;" key then return to MENU"
1850 COM ON
1860 LOCATE 25, LY4
1870 COLOR@(25,LY4)-(25,LY4),6
1880 A$=INKEY$
1890 IF A$="" THEN GOTO 1880
1900 N=2: J=25: LY=LY4
1910 GOSUB 3050
1920 LY4=LY
1930 ON K GOTO 1880,1870,1940,1970
1940 COLOR@(25,LY4)-(25,LY4),0
1950 ON LY4 GOTO 1960,2000
1960 DA$="": GOSUB 2620: GOTO 1870
1970 GOSUB 2960
1980 GOTO 590
1990 COM STOP
2000 LOCATE 25,5: PRINT "M-CH INPUT ""
2010 COLOR@(36,5)-(36,5),6
2020 COM ON
2030 A$=INKEY$
2040 IF A$="" THEN GOTO 2030
2050 IF INSTR ("0123456789"+CHR$(13)+CHR$(28),A$)=0 THEN GOTO 2030
2060 IF A$=CHR$(13) THEN GOTO 2130
2070 IF A$=CHR$(28) THEN GOTO 1970
2080 B1$=B1$+A$
2090 COM STOP
2100 LOCATE 37,5: PRINT B1$
2110 COM ON
```

```
2120 GOTO 2030
2130 C=VAL(B1$)
2-1-40 B1$=""
2150 IF C<100 THEN 2190
2160 C1=INT(C/100)
2170 C=C-100*C1
2180 DA$=CH$(C1)
2190 DA$=DA$+CH$(C)
2200 COLOR@(36,6)-(36,6),0
2210 GOSUB 2620
2220 COM STOP
2230 LOCATE 25,5: PRINT "
2240 COM ON
2250 GOTO 1870
2260 '
2270
           OFFSET WRITE
2280 '
2290 COM STOP
2300 FQ=2
2310 LOCATE 25,4: PRINT "OFFSET FREQ. INPUT (MHz)"
2320 LOCATE 25,8: PRINT ME$;" key then return to MENU"
2330 COLOR@(29,6)-(29,6),6
2340 COM ON
2350 L=8: M=6
2360 GOTO 850
2370
           SCAN CONTROL
2380
2390
2400 COM STOP
2410 LOCATE 25,1 : PRINT "♦ SCAN STOP"
2420 LOCATE 25,2 : PRINT "♦ SCAN START"
2430 LOCATE 25,4 : PRINT ME$;" key then return to MENU"
2440 COM ON
2450 LOCATE 25, LY5
2460 COLOR@(25,LY5)-(25,LY5),6
2470 A$= INKEY$
2480 IF A$="" THEN GOTO 2470
2490 N=2: J=25: LY=LY5
2500 GOSUB 3050
2510 LY5=LY
2520 ON K GOTO 2470,2460,2530,2570
2530 COLOR@(25,LY5)-(25,LY5),0
2540 C=LY5-1
2550 ON LY5 GOTO 2560,2560
2560 DA$=CH$(C): GOSUB 2620: GOTO 2460
2570 GOSUB 2960
2580 GOTO 590
2590
2600
           COMMAND OUTPUT
2610
2620 CNT$=CHR$(LY2-1)
2630 INS$=TX$+CNT$+DA$+FI$
2640 IF LY2<3 THEN 2660
2650 INS$=PR$+RE$+INS$:GOTO 2670
2660 INS$=PR$+CHR$(0)+INS$
2665 LOCATE 3,23 :PRINT "
2670 PRINT#1, INS$;
2680 DA$=""
2690 RETURN
2700
2710
           ON COM
2720
2730 IF LOC(1)=0 THEN RETURN
2740 I$=INPUT$(LOC(1),#1):L=LEN(I$)
2750 FOR K=1 TO L
2760 P_{MID}(I_{K,1}):M=ASC(P_{M})
2770 IF M=&HFC THEN V$="" :GOTO 2810
2780 IF M(16 THEN X$="0"+HEX$(M) ELSE X$=HEX$(M)
2790 IF M=&HFD THEN V$=Z$ :Z$="":X$=""
2800 Z$=X$+Z$
2810 NEXT K
              :IF V$="" THEN RETURN
```

```
2820 L=LEN(V$)
2830 'LOCATE 5,23: PRINT V$
2840 IF INSTR(V$, (RC$+"00FEFE")) <>0 THEN 3460
2850 IF LEFT$(V$,8)=("FA"+RC$+TC$+"FE") THEN LOCATE 5,23 :PRINT "N G"
2860 IF LEFT$(V$,8)=("FB"+RC$+TC$+"FE") THEN LOCATE 5,23 :PRINT "O K"
2870 IF L=32 THEN 3640
2880 S$="03"+RC$+TC$+"FE" : IF INSTR(V$, S$) <>0 THEN 3470
2890 S$="04"+RC$+TC$+"FE" : IF INSTR(V$, S$) <>0 THEN 3540
2900 S$="0C"+RC$+TC$+"FE" : IF INSTR(V$, S$) <>0 THEN 3700
2910 CX=POS(0): CY=CSRLIN
-2920 I $="": RETURN
2930
2940 '
           SUB-ROUTINE CLEAR
2950 '
2960 COM STOP
2970 FOR K=1 TO 11
2980 LOCATE 25.K: PRINT "
2990 NEXT K
3000 COM ON
3010 RETURN
3020 '
3030 '
           SUB-ROUT IN
3040 '
3050 ON INSTR(CHR$(30)+CHR$(31)+CHR$(13)+CHR$(27)+CHR$(28),A$) GOTO 3070,3120,
     3170,3180,3220
3060 K=1: RETURN
3070 COM STOP
3080 LOCATE J, LY: PRINT "♦"
3090 COM ON
3100 IF LY=1 THEN LY=N ELSE LY=LY-1
3110 K=2: RETURN
3120 COM STOP
3130 LOCATE J,LY: PRINT "◆"
3140 COM ON
3150 IF LY=N THEN LY=1 ELSE LY=LY+1
3160 K=2: RETURN
3170 K=3: RETURN
3180 COM STOP
3190 GOSUB 3260
3200 COM ON
3210 K=1: RETURN
3220 K=4: RETURN
3230 '
3240 '
           ADDRESS SET
3250 '
3260 LOCATE 5,2:PRINT "CONTROLLER";
3270 LOCATE 10,3:PRINT "
3280 LOCATE 5,3:INPUT "
                         ADDRESS ",T$
3290 IF T$="" THEN LOCATE 16,3:PRINT HEX$(ASC(TX$));:GOTO 3310
3300 TX$=CHR$(VAL("8H"+T$))
3310 LOCATE 5,5:PRINT "TRANCEIVER";
3320 LOCATE 10,6:PRINT "
3330 LOCATE 5,6: INPUT " ADDRESS ",R$
3340 IF R$="" THEN LOCATE 16,6:PRINT HEX$(ASC(RE$));:GOTO 3360
3350 RE$=CHR$(VAL("&H"+R$))
3360 RC$=RIGHT$(("00"+HEX$(ASC(RE$))),2):TC$=RIGHT$(("00"+HEX$(ASC(TX$))),2)
3370 RETURN
3380 '
3390 '
           STOP
3400 '
3410 CLOSE #1
3420 CONSOLE 0,25,1
3430 CLS
3440 END
3450 *FRQIN
3460 IF L<17 THEN 3540
3470 IF LEFT$(V$,2)="FF" THEN :LOCATE 6,12 :PRINT "-----":GOTO 3520
3480 FQ$=LEFT$(V$,10)
3490 FQ#=VAL(FQ$)/1E+06
3500 LOCATE 6,12
3510 PRINT USING "####.####MHz"; FQ#
```

```
3520 V$="" : RETURN
3530 *MODES
3540 IF L=12 THEN 3600
3550 IF LEFT$(V$,4)="0203" THEN MO$="C W -N" :GOTO 3620
3560 IF LEFT$(V$,4)="0204" THEN MO$="RTTY-N" :GOTO 3620
3570 IF LEFT$(V$,4)="0205" THEN MO$="F M -N" :GOTO 3620
3580 IF LEFT$(V$,4)="0005" THEN MO$="SSB(R)" :GOTO 3620
3590 MO$="--- " :GOTO 3620
3600 C=VAL(LEFT$(V$,2))*4+1 :IF C>35 THEN MO$="---":GOTO 3620
3610 MO$=MID$(MDA$,C,4)
                                   ";: V$=""
3620 LOCATE 16,9:PRINT MO$;"
3630 RETURN
3640 FH$=LEFT$(V$,10):FL$=MID$(V$,13,10)
3650 LOCATE 5,16 :PRINT "UPPER/LOWER FREQUENCY"
3660 FH#=VAL(FH$)/1E+06 :FL#=VAL(FL$)/1E+06
3670 LOCATE 4,17
3680 PRINT USING "####.####MHz -";FH#;:PRINT USING "####.####MHz";FL#
3690 V$="":RETURN
3700 LOCATE 5,19:PRINT "OFFSET FREQUENCY"
3710 OS$=LEFT$(V$,6):F#=VAL(OS$)/10#
3720 LOCATE 6,20:PRINT F#;"KHz
                                      ": V$=""
3730 RETURN
```

## 5. SCHEMATIC DIAGRAM





## **American Communication Systems**

Discover the Power of Communications ™

TO ORDER-VISIT http://www.ameradio.com

## アイコム株式会社